MSP430F1121 – CAN à approximations successives


MSP430F1121

Conversion A/N 4 bits à approximations successives

Objectifs : 

· Apprentissage de l'environnement de développement IAR

· Premier contact avec le jeu d'instruction du MSP430 (on évite les modes d'adressage complexes)

· Simulation et test sur plaque de câblage

· Introduction au langage C : analyse et simulation du même programme en C

Prérequis : 

· Architecture d'un µP

· Notions élémentaires de programmation structurée (sous-programme)

· Algorithme des pesées successives

1. Source assembleur : "CAN_app_succ.s43"

NAME   CAN_app_succ; 

#include  "msp430x11x1.h"  ; 

COL 130

DEBROM EQU 0F000h

DEBRAM EQU 0200h

FINRAM EQU 02FFh

;Start of the main code


     ORG   DEBROM

main       MOV   #WDTPW+WDTHOLD,WDTCTL ;MSB=5Ah Password pour accéder et LSB=80h 

;                                       pour bloquer le chien de garde

           MOV   #FINRAM+1,SP  ;Initialisation du stack pointer

           MOV.b #0FFH,P2DIR   ;P2 en sortie

debut      CALL  #rssi_a       ;\ Boucle de test

           JMP   debut         ;/ 

rssi_a     MOV.b #08H,R5       ;R5: indique le bit en cours de test. On commence avec b3

           CLR.b P2OUT         ;Valeur initiale de P2

new_bita   BIS.b R5,P2OUT      ;R5 or (P2OUT) -> (P2OUT) : forcer le bit suivant à "1"

           BIT.b #1,P1IN       ;Test de la comparaison sur P1.0

           JNE   decal_bit     ;

           XOR.b R5,P2OUT      ;Remise à "0" du bit si dépassement

decal_bit  RRA.b R5            ;Bit suivant à tester

           JEQ   fin_rssi_a    ;EOC ?

           JMP   new_bita      ;Non : on continue

fin_rssi_a MOV.b P2OUT,R4      ;Oui : on place le résultat dans R4

           RET

;Set up the reset vector


     ORG   0FFFEh


     DW    main


     end

Notes :

· Tests sur maquette avec module MSP430F1121 type "OLIMEX"

· CNA 4 bits réalisé avec un réseau R–2R (1K-22K) relié sur P2.3 et P2.0

· Comparateur = LM311 alimenté entre 5V et 0V (Vin sur entrée +)

Fichier "listing" :

###############################################################################

#                                                                             #

#     IAR Systems 430 Assembler V1.26A/WIN KickStart 03/Feb/2003  21:37:46    #

#     Copyright 1996-2002 IAR Systems. All rights reserved.                   #

#                                                                             #

#           Source file   =  C:\MSP430\essai\CAN_app_succ.s43                 #

#           List file     =  c:\msp430\essai\CAN_app_succ.lst                 #

#           Object file   =  c:\msp430\essai\CAN_app_succ.r43                 #

#           Command line  =  -Oc:\msp430\essai\ -s+ -M<> -w+                  #

#                            -Lc:\msp430\essai\ -t8                           #

#                            -IC:\Program Files\IAR Systems\ew23\430\inc\ -r  #

#                            C:\MSP430\essai\CAN_app_succ.s43                 #

#                                                                             #

###############################################################################

      1    00F000              NAME   CAN_app_succ; 

      2    00F000              #include  "msp430x11x1.h"  ; 

      3    00F000                      

      4    00F000              COL 130

      5    00F000              DEBROM     EQU   0F000h

      6    000200              DEBRAM     EQU   0200h

      7    0002FF              FINRAM     EQU   02FFh

      8    00F000              

      9    00F000              ;Start of the main code

     10    000000                         ASEG  CODE

     11    00F000                         ORG   DEBROM

     12    00F000 B040805A1C11 main       MOV   #WDTPW+WDTHOLD,WDTCTL ;MSB=5Ah Password pour accéder et LSB=80

     13    00F006              ;                                       pour bloquer le chien de garde

     14    00F006 31400003                MOV   #FINRAM+1,SP  ;Initialisation du stack pointer

     15    00F00A F0431E10                MOV.b #0FFH,P2DIR   ;P2 en sortie

     16    00F00E B01214F0     debut      CALL  #rssi_a       ;\ Boucle de test

     17    00F012 FD3F                    JMP   debut         ;/ 

     18    00F014 7542         rssi_a     MOV.b #08H,R5       ;R5 : indique le bit en cours de test. On 

     19    00F016 C0431110                CLR.b P2OUT         ;Valeur initiale de P2

     20    00F01A C0D50D10     new_bita   BIS.b R5,P2OUT      ;R5 or (P2OUT) -> (P2OUT) : forcer le bit

     21    00F01E D0B30010                BIT.b #1,P1IN       ;Test de la comparaison sur P1.0

     22    00F022 0220                    JNE   decal_bit     ;

     23    00F024 C0E50310                XOR.b R5,P2OUT      ;Remise à "0" du bit si dépassement

     24    00F028 4511         decal_bit  RRA.b R5            ;Bit suivant à tester

     25    00F02A 0124                    JEQ   fin_rssi_a    ;EOC ?

     26    00F02C F63F                    JMP   new_bita      ;Non : on continue

     27    00F02E 5440F90F     fin_rssi_a MOV.b P2OUT,R4      ;Oui : on place le résultat dans R4

     28    00F032 3041                    RET

     29    00F034                                    

     30    00F034              ; Set up the reset vector

     31    00FFFE                         ORG   0FFFEh

     32    00FFFE 00F0                    DW    main

     33    010000              

     34    010000                         end

##############################

#          CRC:4F02          #

#        Errors:   0         #

#        Warnings: 0         #

#         Bytes: 54          #

##############################

Source C : "CAN_app_succ.c"

//*********************************************************************

//  MSP-FET430F1121- Conversion A/N 4 bits à approximations successives 

//*********************************************************************

#include  <msp430x11x1.h>

char TEST_CAN;

char RSSI_A(void)

{

 char BIT_CAN;

 BIT_CAN = 8;

 P2OUT = 0;

 do

  {

   P2OUT |= BIT_CAN;

   if (P1IN & 1 == 1) P2OUT ^= BIT_CAN;

   BIT_CAN = BIT_CAN >> 1;

  }

 while (BIT_CAN != 0); 

 return P2OUT;

}

void main(void)

{ 

  WDTCTL = WDTPW + WDTHOLD; // Arrêt du chien de garde

  P2DIR  = 0x0F;            // P2.3 à P2.0 en sortie

  for (;;)                              

  {

    TEST_CAN = RSSI_A();

   }  

}

· Présentation de la structure d'un programme "C" : 

· Programme principal : main : lancé au reset. Soit il fait tourner le µP en boucle, soit il le met en veille et attend des interruptions pour limiter la consommation.

· Ensemble de sous-programmes et de programmes d'interruption activés suivant les besoins : exemple RSSI_A,

· Affectation de constantes et de tableaux en mémoire flash.

· Déclaration de variables :

· globales (en RAM, accessibles à tous les SP) ex : "char TEST_CAN" affecte un octet en RAM à la variable TEST_CAN)

· locales (généralement un registre du CPU si la taille le permet) ex : "char BIT_CAN" affecte un registre du CPU à la variable BIT_CAN

· Présentation du source

· Simulation avec l'environnement IAR : on demande de noter :

· le code machine produit par le compilateur C, les adresses ou registres de chaque variable (TEST_CAN, BIT_CAN, P2DIR, P2OUT, P1IN, …) 

· l'adresse du sous-programme RSSI_A

· de comparer le programme "machine" avec le source assembleur

· vérifier l'affectation du vecteur reset

